Структура баз данных СУБД НИМБ.

Сервер1---------∞Таблицы

Таблица1---------∞поля




Таблица1---------∞документы


  1---------∞группы




Документы1--------∞ поля
                           --------∞ группы(группа может содержать поля, одна группа может быть несколько раз)

Группы 1---------∞ поля

Информация о поле 1---------1Тип



          1---------∞ индексы

При запросах можно использовать только индексы.

При запросе Select получаем номер документа, при запросе get document сам документ по его номеру.

Физическое удаление документа из базы не возможно, документ просто помечается как удаленный записью в его метаинформацию.

Формат языка запросов НИМБ 5.2

Размещение текста запроса по строкам может быть произвольным, если это поддерживает язык программирования, из которого делается запрос. В том случае если язык этого не поддерживает, а текст запроса получается слишком длинным и не влезает в одну строку целесообразно присваивать части запроса переменным, а за тем склеиванием строки собирать из них необходимый запрос.

Запросы на получение

1. Запрос на получение имен таблиц.

get table names;

В результате выполнения этого запроса мы получим служебные имена всех таблиц которые присутствуют на данном сервере. Например вот так:

_systemtv

dbcreator_2007_7_9_155653281

dbcreator_2007_9_25_142917312

dbcreator_2007_9_25_15856234

dbcreator_2008_3_27_123751125

dbcreator_2008_3_27_123816703

dbcreator_2008_3_27_12384046

dbcreator_2008_3_27_12391171

dbcreator_2008_4_1_101140843

dbcreator_2008_4_3_132033291

dbcreator_2008_6_27_113041468

dbcreator_2008_6_5_102542703

dbcreator_2008_6_9_11845453

dbcreator_2008_7_14_165154881

dbcreator_2008_7_15_16450906

dbcreator_2008_8_18_181745703

dbcreator_2008_8_20_152622311

dbcreator_2008_8_27_11305631

dbcreator_2008_8_27_192922111

lib2

products.

Число 21 в первой строке показывает общее количество таблиц на данном сервере.  Служебные имена таблиц совпадают с началом названия файлов этих таблиц на диске.

2. Помимо служебного имени у таблиц существуют также имена, которые мы им присвоили (Alias). Например таблица с системным именем dbcreator_2008_4_1_101140843 имеет Alias 

«Информация о доме». Для получения этих имен можно воспользоваться следующим запросом

get table names aliases;

В результате выполнения запроса мы получим ответ в следующем формате. 

…

…

dbcreator_2008_4_1_101140843 alias #17#Информация о доме#

dbcreator_2008_4_3_132033291 alias #38#Товары, услуги и предприятия для диска#

…

…

Число в первой строке выполняет туже функцию что и в предыдущем запросе.  Далее идут служебные имена таблиц до слова alias, а затем уже имя этой таблицы которое мы присвоили ей при создании. Так в строке «dbcreator_2008_4_1_101140843 alias #17#Информация о доме#» служебным именем таблицы будет dbcreator_2008_4_1_101140843, а её назначенным именем будет «Информация о доме». Число 17 между # показывает длину названия таблицы, в символах включая пробелы и знаки препинания.

3. Запрос на получение структуры таблицы.

Каждая таблица имеет свою структуру заданную ей при создании. Структура может состоять из полей и групп.  Внутри группы располагаются вложенные поля, а вот вложенных групп быть не может. Сам запрос будет выглядеть так:

get table <имя таблицы> structure;

Под <имя таблицы> подразумевается её служебное имя.

Пример запроса.

get table dbcreator_2008_4_1_101140843  structure;
Данный запрос вернет следующий результат:

table dbcreator_2008_4_1_101140843 alias #17#Информация о доме# {

   field f1 alias #3#тип# : text null {

      index f1i1 #21#whole size=16 mult=10#;

   };

   field f2 alias #14#название улицы# : text null {

      index f2i1 #21#whole size=16 mult=10#;

   };

   field f3 alias #3#дом# : text null {

      index f3i1 #21#whole size=16 mult=10#;

   };

   field f4 alias #13#Файл описания# : text null {

      index f4i1 #21#whole size=16 mult=10#;

   };

   group g1 alias #10#Фотографии# {

      field g1f1 alias #4#Фото# : file null {

      };

   };

   group g2 alias #17#Информация о доме# {

      field g2f1 alias #6#Баннер# : file null {

      };

      field g2f2 alias #6#Ссылка# : text null {

         index g2f2i1 #21#words size=16 mult=10#;

         index g2f2i2 #21#whole size=16 mult=10#;

      };

   };

};

В первой строке содержится  имя таблицы и её Alias. После символа “{“ описывается непосредственно структура таблицы.

Так надпись“   field f1 alias #3#тип# : text null {

      

    index f1i1 #21#whole size=16 mult=10#;

   

    };  “

Означает что первым в таблице идет поле со служебным именем «f1» и Alias (псевдоним) “тип” цифра 3 между знаками # указывает длину псевдонима в символах. text null означает что поле текстовое, а также может не содержать данных. Далее между символами { и } располагается информация об индексе этого поля. Его имя и множителя слияния. Например, whole size whole — тип индекса (значение целиком), size — размер "кластера" индекса, mult – множитель слияния. Аналогичным образом описаны остальные поля и группы.

4. Запрос на получение длины таблицы.

get table <имя таблицы> length;
Пример запроса:

get table dbcreator_2008_4_1_101140843 length;
В качестве ответа сервер выведет два числа. Первое обозначает количество актуальных документов в таблице, а второе общее количество документов.

5.  Запрос на получение списка документов по параметрам.

select [including updated или only updated] from <имя таблицы> where <выражение>;

including updated помещена в квадратные скобки так как может присутствовать в данном запросе а может и нет. В зависимости от того,  что мы хотим получить в результате запроса. Если мы выполним запрос в  виде:

select including updated from <имя таблицы> where <выражение>;
то получим только список  не актуальных документов. В противном случае мы получим список всех документов. Под <имя таблицы> подразумевается служебное имя таблицы. Под <выражение> подразумевается логическое выражение отношения параметров по которым мы выбираем документ.

<выражение> = <логическое слагаемое> {or <логическое слагаемое>}.

<логическое слагаемое> = <логический множитель> {and <логический множитель>}.

<логический множитель> = <элементарное сравнение> или <сравнение с диапазоном> или "(" <выражение ")".

<элементарное сравнение> = [<имя группы> "."] <имя поля> "." <имя индекса> <отношение> <строка с решётками>.

<отношение> = "=" | "<>" | "<=" | ">=" | "<" | ">" | prefix.

<сравнение с диапазоном> = [<имя группы> "."] <имя поля> "." <имя индекса> range <строка с решётками> to <строка с решётками>.

Например
select from table1 
where field1.index1 = #6#value1# or group1.field2.index2 

      < #6#value2# or group2.field3.index3 prefix #1#a# and group2.field3.index3 

      prefix #1#б#;

выбрать из таблицы table1 документы, в которых в индексе index1 поля field1 

есть значение value1 или в индексе index2 поля field2 группы group1 есть 

значения, меньшие value2 или индекс index3 поля field3 группы group2 

содержит строки, начинающиеся на а и на б (это одновременно возможно, если 

от одного поля в индекс попадает несколько значений, скажем, при 

индексировании по словам, а также возможно, поскольку экземпляров групп 

может быть несколько).
При этом ответ запроса будет выглядеть следующим образом: сначала будет число показывающее общее количество найденных документов, а затем номера этих документов в формате Dword (4 байта). 

6. Запрос на получение метаинформации документа.

get document <номер документа> from table <имя таблицы> meta ;

Пример запроса:

get document 12 from table dbcreator_2008_4_1_101140843 meta ;

Ответ на запрос приходит в следующем формате:

document 
{ 

usercreated = #5#admin#; 

creationtime = #19#01.04.2008 06:12:01#; 

replaced = #4#none#; 

replacedby = #3#480#; 
};
Значение параметра usercreated показывает кто из пользователей создал документ, значение creationtime показывает когда был создан этот документ (время создания по Гринвичу). Значение replaced показывает заменяет ли этот документ какой-нибудь другой, а значение replacedby заменен ли этот документ каким-нибудь  другим.

7. Запрос на получение документа по номеру

get {document <номер документа> from table <имя таблицы>

  [only <имя группы или поля> {"," <имя группы или поля>}] ";"}.

Примет запроса:

Get document 15 from table dbcreator_2008_4_1_101140843; 

Ответ приходит в следующем формате:

document { field f1 = #5#улица#; field f2 = #12#Воркутинская#; field f3 = #1#1#; field f4 = null; group g1 { field g1f1 = #28507#File-Path: .\Воркутинская 1.jpg

Структура ответа аналогична описанным выше. запросы на получение метаинформации и на получение документа по номеру могут идти под одним get в произвольном порядке

8. запрос на получение количества ключей индекса

get index <имя таблицы> "." [<имя группы> "."] <имя поля> "." <имя индекса> length ";".

Пример запроса:

Get index dbcreator_2008_4_1_101140843.f1.f1i1 length;

В качестве ответа на запрос приходит просто число. Этот запрос требует прохода по всему индексу, поэтому если индекс большой — использовать его не рекомендуется.
9. запрос на получение интервала ключей индекса

get index <имя таблицы> "." [<имя группы> "."] <имя поля> "." <имя индекса>

   from (<десятичное неотрицательное число> или key <строка в решётках>)

   length <десятичное неотрицательное число> ";"

   // from 0 - с начала

Пример запроса:

Get index dbcreator_2008_4_1_101140843.f1.f1i1 from 0 length 20;

Ответ приходит в виде:

#10#МИКРОРАЙОН#
#8#ПЕРЕУЛОК#
#7#ПЛОЩАДЬ#
#5#УЛИЦА#

Запросы на добавление, изменение.

1. Запрос на создание таблицы. 

create table <имя таблицы> alias <строка с решётками> "{"

{<описание поля>|<описание группы>}

"}" ";".

<описание группы> =

group <имя группы> alias <строка с решётками> "{"

{<описание поля>}

"}" ";".

<описание поля> =

field <имя поля> alias <строка с решётками> ":" <тип поля> "{"

{<описание индекса>}

"}" ";".

<тип поля> = text | int64 | double | date | time | datetime | binary.

<описание индекса> =

index <имя индекса> <строка с решётками> ";".

// "whole" - целиком, "words" - по словам,

// допускается приписка про размер кластера, например,

// "whole size=16", допустимы числа от 16 до 24.

Пример запроса:

Create table primer alias #15#Пример_создания# {field f1 alias #5#nomer#:text{index i1 #5#words#;};};

2.  Запрос на удаление таблицы.

drop table <имя таблицы> ";"

Пример запроса:

drop table primer;

В качестве имени таблицы нужно указывать служебное имя, а не alias.

3. Запрос на добавление замену документа

add document "{"

{<данные поля>|<данные группы>}

"}" to table <имя таблицы> [replacing <номер документа>]";".

// replacing:

 в результате старый документ остаётся под тем же номером,

 помечается как устаревший, в нём появляется ссылка на новый

 документ, а в новом документе появляется ссылка на старый

 изменяемый документ не должен быть устаревшим

<данные группы> =

group <имя группы> "{"

{<данные поля>}

"}" ";".

<данные поля> =

field <имя поля> "=" <строка с решётками> ";".

Пример запроса:

add document {

   field f_one = #6#поле 1#;

   field f_two = #6#поле 2#;

   group one {

      field one = #17#группа 1 поле 1 а#;

   };

   group one {

      field one = #17#группа 1 поле 1 б#;

   };

   group two {

      field f_one = #17#группа 2 поле 1 а#;

      field f_two = #17#группа 2 поле 2 а#;

   };

   group two {

      field f_one = #17#группа 2 поле 1 б#;

      field f_two = #17#группа 2 поле 2 б#;

   };

} to table g2;

В данном примере в таблицу g2 добавляется документ с описанными полями и группами.

Прочие запросы

1. Запрос на фиксацию состояния сервера.

commit;

